

TRAVE\_MISTA

```

finish
/clear

c***
c*** trave incastrata modellata con tipi di elemento diversi
c***

/title,Trave incastrata modellata con tipi di elemento diversi

! parametri di geometria
pi = acos(-1)
de = 40          !diametro esterno
di = 20          !diametro interno
dm = (de+di)/2  !diametro medio
tk = (de-di)/2  !spessore
l = 390          !lunghezza

! parametri di carico
p = 200          !forza all'estremo libero

! parametri di materiale
e=205000
nu=0.3

/shade,,0

! parametri di suddivisione
! suddivisione del quarto di circonferenza
DivQ=10
! suddivisione in senso radiale
DivR=4
! suddivisioni in senso longitudinale
DivL=12

/prep7

! definizione tipi di elemento
! elemento brick
et,1,185
! elemento shell
et,2,181
! elemento beam
et,3,188,,3 !shape function cubica

! definizione del materiale
mp,ex,1,E
mp,prxy,1,nu

! definizione del tipo di sezione
sectype,1,shell
secdata,tk
sectype,2,beam,ctube
secdata,di/2,de/2,DivQ*4 !raggio interno, esterno, numero celle circolari

! modellazione primo tratto
cyl4,0,0,di/2,0,de/2,90,L/3 !CYL4, XCENTER, YCENTER, RAD1, THETA1, RAD2, THETA2,
DEPTH
cyl4,0,0,di/2,90,de/2,180,L/3
cyl4,0,0,di/2,180,de/2,270,L/3
cyl4,0,0,di/2,270,de/2,360,L/3
vglue,all

! imposizione delle suddivisioni delle linee
lsel,s,loc,z,-0.01,0.01
lsel,a,loc,z,L/3-0.01,L/3+0.01
csys,1
lsel,r,loc,x,di/2+0.01,de/2-0.01
csys,0

```

TRAVE\_MISTA

```

lesize,all,,,DivR
lsel,s,loc,z,-0.01,0.01
lsel,a,loc,z,L/3-0.01,L/3+0.01
csys,1
lsel,r,loc,x,de/2-0.01,de/2+0.01
!lsel,a,loc,x,di/2-0.01,di/2+0.01
csys,0
lesize,all,,,DivQ
lsel,s,loc,z,0.01,L/3-0.01
lesize,all,,,DivL
lsel,all

! mesh del tratto con elementi brick
type,1
vsweep,all !fills a volume with elements by sweeping the mesh from an adjacent
area

! modellazione secondo tratto
k,101,0,0,L/3 !coordinate estremi asse tubo
k,102,0,0,2*L/3 !
k,103,Dm/2,0,L/3 !punto di asse circonferenza

circle,101,Dm/2,102,103 !CIRCLE, PCENT, RAD, PAXI circonferenza di base
l,101,102 !asse tubo
lsel,s,loc,z,L/3+0.01,2*L/3-0.01 !selezione asse
*get,num1,line,0,num,max ! trovo il numero della linea d'asse
lsel,s,loc,z,L/3-0.01,L/3+0.01 !linee sulla base
csys,1
lsel,r,loc,x,Dm/2-0.01,Dm/2+0.01 !linee con centro su circonferenza mediana
csys,0
lsel,u,loc,x,-0.01,0.01 !escludo linee a croce
lsel,u,loc,y,-0.01,0.01

adrag,all,,,,,num1 !superficie mediana tubo generata per estrusione

! imposizione delle suddivisioni
lsel,s,loc,z,2*L/3-0.01,2*L/3+0.01 !su circonferenza
lesize,all,,,DivQ
lsel,s,loc,z,L/3+0.01,2*L/3-0.01 ! su generatrici
lesize,all,,,DivL
alls

! mesh del tratto con elementi shell
type,2
secnum,1

asel,s,loc,z,L/3+0.01,2*L/3-0.01 !seleziona area in base alla posizione centro
mshkey,1
amesh,all

! modellazione terzo tratto
k,201,0,0,2*L/3
k,202,0,0,L

l,201,202
lsel,s,loc,z,2*L/3+0.01,L-0.01 !selezione asse
*get,num2,line,0,num,max ! trovo il numero della linea d'asse

lesize,num2,,,1 !una sola divisione

! mesh del tratto con elementi pipe
type,3
secnum,2
lmesh,num2

!/eshape,1 !se si vuole rappresentare con la vera forma

```

```

                                TRAVE_MISTA
! applicazione vincoli CERIG fra shell e solid
local,11,0,0,0,1/3 !cs traslato rispetto al globale di L/3
clocal,13,1,0,0,0 !cs cilindrico rispetto al local
*do,i,1,4*DivQ
    ! selezione del nodo master per ogni settore
    esel,s,type,,2
    nsel,s
    NodoMaster=node(Dm/2,(i-1)/(4*DivQ)*360,0.0)
    nsel,s,loc,y,(i-1)/(4*DivQ)*360-0.01,(i-1)/(4*DivQ)*360+0.01
    nsel,r,loc,z,-0.01,+0.01
    cerig,NodoMaster,all,all !CERIG, MASTE, SLAVE, Ldof,
    alls
*enddo

! applicazione del CERIG fra shell e pipe
clocal,12,0,0,0,1/3 !cs traslato rispetto al local di L/3
! individuazione del nodo MASTER
NodoMaster=node(0,0,0)
! selezione dei nodi slave
nsel,s,loc,z,-0.01,+0.01

cerig,NodoMaster,all,all

alls

! vincoli e carichi
csys
! incastro ad un'estremità
nsel,s,loc,z,-0.01,+0.01
d,all,ux
d,all,uy
d,all,uz
alls

! carico concentrato all'altra
nsel,s,loc,z,L-0.01,L+0.01
f,all,fy,-P
alls
finish

! soluzione
/solu
solve
finish

! post-processing
/post1
plnsol,s,z

```